

## NAG Library Routine Document

### F07MBF (DSYSVX)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

F07MBF (DSYSVX) uses the diagonal pivoting factorization to compute the solution to a real system of linear equations

$$AX = B,$$

where  $A$  is an  $n$  by  $n$  symmetric matrix and  $X$  and  $B$  are  $n$  by  $r$  matrices. Error bounds on the solution and a condition estimate are also provided.

#### 2 Specification

```

SUBROUTINE F07MBF (FACT, UPLO, N, NRHS, A, LDA, AF, LDAF, IPIV, B, LDB,      &
                  X, LDX, RCOND, FERR, BERR, WORK, LWORK, IWORK, INFO)
INTEGER           N, NRHS, LDA, LDAF, IPIV(*), LDB, LDX, LWORK,          &
                  IWORK(*), INFO
REAL (KIND=nag_wp) A(LDA,*), AF(LDAF,*), B(LDB,*), X(LDX,*), RCOND,      &
                  FERR(*), BERR(*), WORK(max(1,LWORK))
CHARACTER(1)     FACT, UPLO

```

The routine may be called by its LAPACK name *dsysvx*.

#### 3 Description

F07MBF (DSYSVX) performs the following steps:

1. If FACT = 'N', the diagonal pivoting method is used to factor  $A$ . The form of the factorization is  $A = UDU^T$  if UPLO = 'U' or  $A = LDL^T$  if UPLO = 'L', where  $U$  (or  $L$ ) is a product of permutation and unit upper (lower) triangular matrices, and  $D$  is symmetric and block diagonal with 1 by 1 and 2 by 2 diagonal blocks.
2. If some  $d_{ii} = 0$ , so that  $D$  is exactly singular, then the routine returns with INFO =  $i$ . Otherwise, the factored form of  $A$  is used to estimate the condition number of the matrix  $A$ . If the reciprocal of the condition number is less than *machine precision*, INFO =  $N + 1$  is returned as a warning, but the routine still goes on to solve for  $X$  and compute error bounds as described below.
3. The system of equations is solved for  $X$  using the factored form of  $A$ .
4. Iterative refinement is applied to improve the computed solution matrix and to calculate error bounds and backward error estimates for it.

#### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

## 5 Arguments

- 1: FACT – CHARACTER(1) *Input*  
*On entry:* specifies whether or not the factorized form of the matrix  $A$  has been supplied.  
 FACT = 'F'  
 AF and IPIV contain the factorized form of the matrix  $A$ . AF and IPIV will not be modified.  
 FACT = 'N'  
 The matrix  $A$  will be copied to AF and factorized.  
*Constraint:* FACT = 'F' or 'N'.
- 2: UPLO – CHARACTER(1) *Input*  
*On entry:* if UPLO = 'U', the upper triangle of  $A$  is stored.  
 If UPLO = 'L', the lower triangle of  $A$  is stored.  
*Constraint:* UPLO = 'U' or 'L'.
- 3: N – INTEGER *Input*  
*On entry:*  $n$ , the number of linear equations, i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 4: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides, i.e., the number of columns of the matrix  $B$ .  
*Constraint:* NRHS  $\geq 0$ .
- 5: A(LDA,\*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  symmetric matrix  $A$ .  
 If UPLO = 'U', the upper triangular part of  $A$  must be stored and the elements of the array below the diagonal are not referenced.  
 If UPLO = 'L', the lower triangular part of  $A$  must be stored and the elements of the array above the diagonal are not referenced.
- 6: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F07MBF (DSYSVX) is called.  
*Constraint:* LDA  $\geq \max(1, N)$ .
- 7: AF(LDAF,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array AF must be at least  $\max(1, N)$ .  
*On entry:* if FACT = 'F', AF contains the block diagonal matrix  $D$  and the multipliers used to obtain the factor  $U$  or  $L$  from the factorization  $A = UDU^T$  or  $A = LDL^T$  as computed by F07MDF (DSYTRF).  
*On exit:* if FACT = 'N', AF returns the block diagonal matrix  $D$  and the multipliers used to obtain the factor  $U$  or  $L$  from the factorization  $A = UDU^T$  or  $A = LDL^T$ .

- 8: LDAF – INTEGER *Input*  
*On entry:* the first dimension of the array AF as declared in the (sub)program from which F07MBF (DSYSVX) is called.  
*Constraint:* LDAF  $\geq \max(1, N)$ .
- 9: IPIV(\*) – INTEGER array *Input/Output*  
**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .  
*On entry:* if FACT = 'F', IPIV contains details of the interchanges and the block structure of  $D$ , as determined by F07MDF (DSYTRF).  
 if  $IPIV(i) = k > 0$ ,  $d_{ii}$  is a 1 by 1 pivot block and the  $i$ th row and column of  $A$  were interchanged with the  $k$ th row and column;  
 if UPLO = 'U' and  $IPIV(i - 1) = IPIV(i) = -l < 0$ ,  $\begin{pmatrix} d_{i-1,i-1} & \bar{d}_{i,i-1} \\ \bar{d}_{i,i-1} & d_{ii} \end{pmatrix}$  is a 2 by 2 pivot block and the  $(i - 1)$ th row and column of  $A$  were interchanged with the  $l$ th row and column;  
 if UPLO = 'L' and  $IPIV(i) = IPIV(i + 1) = -m < 0$ ,  $\begin{pmatrix} d_{ii} & d_{i+1,i} \\ d_{i+1,i} & d_{i+1,i+1} \end{pmatrix}$  is a 2 by 2 pivot block and the  $(i + 1)$ th row and column of  $A$  were interchanged with the  $m$ th row and column.  
*On exit:* if FACT = 'N', IPIV contains details of the interchanges and the block structure of  $D$ , as determined by F07MDF (DSYTRF), as described above.
- 10: B(LDB, \*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the second dimension of the array B must be at least  $\max(1, NRHS)$ .  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .
- 11: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F07MBF (DSYSVX) is called.  
*Constraint:* LDB  $\geq \max(1, N)$ .
- 12: X(LDX, \*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the second dimension of the array X must be at least  $\max(1, NRHS)$ .  
*On exit:* if INFO = 0 or  $N + 1$ , the  $n$  by  $r$  solution matrix  $X$ .
- 13: LDX – INTEGER *Input*  
*On entry:* the first dimension of the array X as declared in the (sub)program from which F07MBF (DSYSVX) is called.  
*Constraint:* LDX  $\geq \max(1, N)$ .
- 14: RCOND – REAL (KIND=nag\_wp) *Output*  
*On exit:* the estimate of the reciprocal condition number of the matrix  $A$ . If RCOND = 0.0, the matrix may be exactly singular. This condition is indicated by INFO  $> 0$  and INFO  $\leq N$ . Otherwise, if RCOND is less than the *machine precision*, the matrix is singular to working precision. This condition is indicated by INFO =  $N + 1$ .

15: FERR(\*) – REAL (KIND=nag\_wp) array Output

**Note:** the dimension of the array FERR must be at least  $\max(1, \text{NRHS})$ .

*On exit:* if  $\text{INFO} = 0$  or  $N + 1$ , an estimate of the forward error bound for each computed solution vector, such that  $\|\hat{x}_j - x_j\|_\infty / \|x_j\|_\infty \leq \text{FERR}(j)$  where  $\hat{x}_j$  is the  $j$ th column of the computed solution returned in the array **X** and  $x_j$  is the corresponding column of the exact solution  $X$ . The estimate is as reliable as the estimate for **RCOND**, and is almost always a slight overestimate of the true error.

16: BERR(\*) – REAL (KIND=nag\_wp) array Output

**Note:** the dimension of the array BERR must be at least  $\max(1, \text{NRHS})$ .

*On exit:* if  $\text{INFO} = 0$  or  $N + 1$ , an estimate of the component-wise relative backward error of each computed solution vector  $\hat{x}_j$  (i.e., the smallest relative change in any element of  $A$  or  $B$  that makes  $\hat{x}_j$  an exact solution).

17: WORK(max(1, LWORK)) – REAL (KIND=nag\_wp) array Workspace

*On exit:* if  $\text{INFO} = 0$ , **WORK**(1) returns the optimal **LWORK**.

18: LWORK – INTEGER Input

*On entry:* the dimension of the array **WORK** as declared in the (sub)program from which **F07MBF** (**DSYSVX**) is called.

$\text{LWORK} \geq \max(1, 3 \times N)$ , and for best performance, when **FACT** = 'N',  $\text{LWORK} \geq \max(1, 3 \times N, N \times nb)$ , where  $nb$  is the optimal block size for **F07MDF** (**DSYTRF**).

If  $\text{LWORK} = -1$ , a workspace query is assumed; the routine only calculates the optimal size of the **WORK** array, returns this value as the first entry of the **WORK** array, and no error message related to **LWORK** is issued.

19: IWORK(\*) – INTEGER array Workspace

**Note:** the dimension of the array **IWORK** must be at least  $\max(1, N)$ .

20: INFO – INTEGER Output

*On exit:*  $\text{INFO} = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

$\text{INFO} < 0$

If  $\text{INFO} = -i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$  and  $\text{INFO} \leq N$

Element  $\langle \text{value} \rangle$  of the diagonal is exactly zero. The factorization has been completed, but the factor  $D$  is exactly singular, so the solution and error bounds could not be computed.  $\text{RCOND} = 0.0$  is returned.

$\text{INFO} = N + 1$

$D$  is nonsingular, but  $\text{RCOND}$  is less than *machine precision*, meaning that the matrix is singular to working precision. Nevertheless, the solution and error bounds are computed because there are a number of situations where the computed solution can be more accurate than the value of  $\text{RCOND}$  would suggest.

## 7 Accuracy

For each right-hand side vector  $b$ , the computed solution  $\hat{x}$  is the exact solution of a perturbed system of equations  $(A + E)\hat{x} = b$ , where

$$\|E\|_1 = O(\epsilon)\|A\|_1,$$

where  $\epsilon$  is the *machine precision*. See Chapter 11 of Higham (2002) for further details.

If  $\hat{x}$  is the true solution, then the computed solution  $x$  satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|\hat{x}\|_\infty} \leq w_c \text{cond}(A, \hat{x}, b)$$

where  $\text{cond}(A, \hat{x}, b) = \frac{\| |A^{-1}|(|A|\|\hat{x}\| + |b|) \|_\infty}{\|\hat{x}\|_\infty} \leq \text{cond}(A) = \| |A^{-1}| |A| \|_\infty \leq \kappa_\infty(A)$ . If  $\hat{x}$  is the  $j$ th column of  $X$ , then  $w_c$  is returned in `BERR(j)` and a bound on  $\|x - \hat{x}\|_\infty / \|\hat{x}\|_\infty$  is returned in `FERR(j)`. See Section 4.4 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F07MBF (DSYSVX) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07MBF (DSYSVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The factorization of  $A$  requires approximately  $\frac{1}{3}n^3$  floating-point operations.

For each right-hand side, computation of the backward error involves a minimum of  $4n^2$  floating-point operations. Each step of iterative refinement involves an additional  $6n^2$  operations. At most five steps of iterative refinement are performed, but usually only one or two steps are required. Estimating the forward error involves solving a number of systems of equations of the form  $Ax = b$ ; the number is usually 4 or 5 and never more than 11. Each solution involves approximately  $2n^2$  operations.

The complex analogues of this routine are F07MPF (ZHESVX) for Hermitian matrices, and F07NPF (ZSYSVX) for symmetric matrices.

## 10 Example

This example solves the equations

$$AX = B,$$

where  $A$  is the symmetric matrix

$$A = \begin{pmatrix} -1.81 & 2.06 & 0.63 & -1.15 \\ 2.06 & 1.15 & 1.87 & 4.20 \\ 0.63 & 1.87 & -0.21 & 3.87 \\ -1.15 & 4.20 & 3.87 & 2.07 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 0.96 & 3.93 \\ 6.07 & 19.25 \\ 8.38 & 9.90 \\ 9.50 & 27.85 \end{pmatrix}.$$

Error estimates for the solutions, and an estimate of the reciprocal of the condition number of the matrix  $A$  are also output.

## 10.1 Program Text

```

Program f07mbfe

!      F07MBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: dsysvx, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nb = 64, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: rcond
Integer                    :: i, ifail, info, lda, ldaf, ldb, ldx, &
                          lwork, n, nrhs
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,,:), af(:,,:), b(:,,:), berr(:), &
                          ferr(:), work(:), x(:,,:)
Integer, Allocatable        :: ipiv(:), iwork(:)
!      .. Executable Statements ..
Write (nout,*) 'F07MBF Example Program Results'
Write (nout,*)
Flush (nout)
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs
lda = n
ldaf = n
ldb = n
ldx = n
lwork = nb*n
Allocate (a(lda,n),af(ldaf,n),b(ldb,nrhs),berr(nrhs),ferr(nrhs), &
        work(lwork),x(ldx,nrhs),ipiv(n),iwork(n))

!      Read the upper triangular part of A from data file

Read (nin,*)(a(i,i:n),i=1,n)

!      Read B from data file

Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Solve the equations AX = B for X
!      The NAG name equivalent of dsysvx is f07mbf
Call dsysvx('Not factored','Upper',n,nrhs,a,lda,af,ldaf,ipiv,b,ldb,x, &
        ldx,rcond,ferr,berr,work,lwork,iwork,info)

If ((info==0) .Or. (info==n+1)) Then

!      Print solution, error bounds and condition number

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04caf('General',' ',n,nrhs,x,ldx,'Solution(s)',ifail)

Write (nout,*)
Write (nout,*) 'Backward errors (machine-dependent)'
Write (nout,99999) berr(1:nrhs)
Write (nout,*)
Write (nout,*) 'Estimated forward error bounds (machine-dependent)'
Write (nout,99999) ferr(1:nrhs)
Write (nout,*)
Write (nout,*) 'Estimate of reciprocal condition number'
Write (nout,99999) rcond
Write (nout,*)

If (info==n+1) Then

```

```

        Write (nout,*)
        Write (nout,*) 'The matrix A is singular to working precision'
    End If
Else
    Write (nout,99998) 'The diagonal block ', info, ' of D is zero'
End If

99999 Format ((3X,1P,7E11.1))
99998 Format (1X,A,I3,A)
End Program f07mbfe

```

## 10.2 Program Data

```

F07MBF Example Program Data
  4      2      :Values of N and NRHS
-1.81   2.06   0.63  -1.15
        1.15   1.87   4.20
                -0.21  3.87
                2.07 :End of matrix A

0.96   3.93
6.07  19.25
8.38   9.90
9.50  27.85      :End of matrix B

```

## 10.3 Program Results

F07MBF Example Program Results

Solution(s)

	1	2
1	-5.0000	2.0000
2	-2.0000	3.0000
3	1.0000	4.0000
4	4.0000	1.0000

Backward errors (machine-dependent)

1.4E-16	1.0E-16
---------	---------

Estimated forward error bounds (machine-dependent)

2.5E-14	3.2E-14
---------	---------

Estimate of reciprocal condition number

1.3E-02
---------